

3.2. Valószínűségek számítása geometriai és számítógép algebrai eszközökkel

Valószínűségszámítás és statisztika számítógép algebrai támogatással	Készült a PTE PMMIK-n a TÁMOP -4.1.2.A/1-11/1-2011-0098 támogatásával
Klincsik Mihály	

3.2.1 Bevezetés

Ha egy véletlen kísérlet Ω eseménytérét geometriai alakzattal tudjuk leírni, gondolunk itt egy szakasz, háromszög, téglalap, kör, téglatest, gömb stb. típusú alakzatokra, akkor a tér dimenziójától függően valamely $A \subseteq \Omega$ részhalmaz $m(A)$ mértékét a **hossz**, **terület** illetve **térfogat** adja meg.

Az $A \subseteq \Omega$ esemény valószínűségét klasszikus valószínűségi mezőben a

$$P(A) = \frac{m(A)}{m(\Omega)}$$

mértékek hányadosával értelmezzük, ahol $m(A)$ az A esemény mértéke és $m(\Omega)$ az eseménytér mértéke. Az így értelmezett valószínűség mindig 0 és 1 közé esik, mivel $0 \leq m(A) \leq m(\Omega)$, hiszen kisebb halmaz mértéke kisebb. Továbbá $P(A)$ teljesíti a valószínűségekre tett σ -additív tulajdonságot is.

Mivel **geometriai alakzatok** fordulnak elő a valószínűségek ilyen számítása során, ezért kapta a módszer a geometriai valószínűség nevet.

Emlékeztetünk arra, hogy a **hosszúság**, a **terület** és a **térfogat** kiszámításához bizonyos esetekben szükségesek a differenciál és **integrálszámítás** eszközei is. Ezért a valószínűségelmélet ezen részében felhasználjuk az integrálszámítás eredményeit.

A mostani számításokat részben már tárgyaltuk az 1. fejezet 1.7 pontjában az eseménytér konstrukciójánál és a 2. fejezet 2.7 és 2.8 pontjaiban a valószínűségek meghatározásánál.

Célunk az alábbi fogalmak, összefüggések és eljárások megismerése, megértése

Folytonos Ω eseménytér meghatározása feladatok szövege alapján

Folytonos egyenletes eloszlás megadása a **Statistics** csomag *Uniform(a, b)* segítségével

Folytonos Ω eseménytér dimenziójának és $m(\Omega)$ mértékének számítása.

Maple **Statistics** csomag
RandomVariable eljárásának alkalmazása folytonos változók létrehozására

Az A esemény berajzolása a folytonos Ω eseménytérbe.

Műveletek folytonos valószínűségi változókkal

Az A esemény $m(A)$ mértékének kiszámítása.

A **Statistics** csomag *Probability* eljárásának alkalmazása

Az A esemény $P(A)$ valószínűségének számítása $\frac{m(A)}{m(\Omega)}$ hányados alapján.

Folytonos valószínűségi változókra az
• eloszlás függvény számítása a **Statistics** csomag *CDF(X, x)* eljárásával
• sűrűség függvény számítása a **Statistics** csomag *PDF(X, x)* eljárásával

Szimulációs Maple program írása

A **plots** csomag *inequal* eljárásával lineáris

véletlen számok generálásával folytonos eloszlással egyenlőtlenségekkel határolt sík tartományok ábrázolása

Az 1D-s és 2D-s folytonos eseményterek megkülönböztetése

3.2.2. Valószínűségek számítása folytonos eseménytérben

Milyen változásokkal kell számolnunk a valószínűségek meghatározásánál, ha az eseménytér amelyben dolgozunk nem diszkrét, hanem folytonos. A véges diszkrét terek esetében Maple segítségével általában elő tudtuk állítani az eseménytér elemeit és a megfelelő részhalmazt megszámláltattuk egy programmal. Ez nem működik folytonos esetben. Nagyobb szerepet kap a szemléltetés.

Egészítsük ki azon tevékenységek táblázatát, amelyet 2.7. pontban a valószínűségek kiszámításánál adtunk a folytonos eseménytérre.

Tevékenységek geometriai valószínűségek számítása során	
1. lépés.	A folytonos Ω eseménytér kiválasztása, szemléltetése és $m(\Omega)$ mértékének számítása
2. lépés.	A feladatban szereplő E esemény meghatározása és szemléltetése az eseménytérben
3. lépés.	Az E esemény $m(E)$ mértékének számítása
4. lépés.	A $P(E) = \frac{m(E)}{m(\Omega)}$ hányadossal számoljuk az E esemény valószínűségét.
5. lépés.	A feladat szimulációja
6. lépés.	A <i>Statistics</i> csomag segítségével próbáljuk automatikusan számolni a valószínűséget.

A szemléltetés, a szimuláció és a véletlen változók, azok az elemek, amelyek újként kerültek a táblázatba.

Fontos az, hogy az eseménytér dimenzió számát a feladatnak megfelelően helyesen állapítsuk meg. Találkoztunk (név nélkül) ugyanis olyan tankönyvvel, amelyben helytelenül állapították meg az eseményteret. Kettő dimenziós helyet egy dimenziós eseményteret adtak meg. Ilyen csapdába könnyen beleeshetünk.

3.2.3. Valószínűségek számítása 1D-s geometriai eszközökkel

A következő példában az **eseménytér** egy intervallum lesz és az **eseményalgebra** a részintervallumok σ -algebrája. Arra építjük a valószínűségek számítását, hogy a **valószínűségi mérték** az intervallum **hosszával** arányos.

3.2.3.1. PÉLDA. 1D-s (Ω, S, P) valószínűségi mező egyenletes eloszlással

Osszuk két részre a $[0,1]$ intervallumot egy véletlenszerűen választott X számmal.

(a) Mekkora **valószínűséggel** lesz a keletkezett két rész négyzetének összege nagyobb, mint $\frac{5}{9}$?

(b) Generáljunk a $[0,1]$ intervallumban 500 darab x_n pontot egyenletes eloszlással, majd

rajzoljuk fel a

$$y_n = x_n^2 + (1 - x_n)^2 > \frac{5}{9}$$

feltételt teljesítő pontok **relatív gyakoriságait**! Vessük össze a kapott ábrát az (a) pontban kapott valószínűséggel!

(c) A *Statistics* csomag segítségével állítsuk elő az $Y = X^2 + (1 - X)^2$ véletlen változót és számoljuk ki a $P\left(Y > \frac{5}{9}\right)$ valószínűséget! Közelítsük az $(y_n, n = 1 \dots 500)$ pontok **gyakoriság hisztogramját** Y **sűrűség függvényével**!

Megoldás

Mi a **véletlen kísérlet** a példában? Az X pont véletlenszerű választása $[0,1]$ -ben. Az **eseménytér** a kísérlet összes lehetséges **kimenetele**inek halmaza és ez a $[0,1]$ intervallum $\Omega = [0, 1]$.

Vigyázat ne tévesszen meg senkit a kimenetel szó! Itt a kimenetelt nem az $x^2 + (1 - x)^2$ formula jelenti és Ω **nem** ennek összes lehetséges értéke, mert akkor helytelenül $\left[\frac{1}{2}, 1\right]$ lenne az eseménytér!

A példa az $\Omega = [0, 1]$ intervallum pontjaiból táplálkozik, amelynek számossága (nem megszámlálhatóan) végtelen. Ezért $m(\Omega) = 1$ az intervallum hossza méri a benne levő pontok mennyiségét.

(a) Meg kell határozni a $[0, 1]$ intervallumon belül a **kedvező esetek**

$$A = \left\{ X \in \Omega : X^2 + (1 - X)^2 > \frac{5}{9} \right\}$$

részalmazát és ennek $m(A)$ **mértékét** vagy inkább összhosszát! A mértékek hányadosa adja az A esemény **$P(A)$ valószínűségét**. Nevezetesen $P(A)$ egyenlő a kedvező esetek mértéke és az összes esetek mértékének hányadosa, ahol **1 dimenzióban** a mérték jelentése **hosszúság**


$$P(A) = \frac{m(A)}{m(\Omega)} = m(A).$$

Az A halmazt azon x -pontok alkotják a $[0,1]$ intervallumban, amelyekre az $y = x^2 + (1 - x)^2$ parabola értéke az $y = \frac{5}{9}$ konstans szint fölött van. Rajzoljuk fel a két függvényt!

Az $y = x^2 + (1 - x)^2$ parabola és az $y = \frac{5}{9}$ konstans függvény

> restart :

```
rajz1 := plot([x^2 + (1 - x)^2, 5/9], x = 0..1, 0..1, thickness = 3, color = [red, blue],  
scaling = constrained) : rajz1
```



Látható, hogy az $y = x^2 + (1 - x)^2$ parabola kettő x -helyen is felveszi az $y = \frac{5}{9}$ konstans értéket. Ezeket `solve` parancs segítségével kapjuk meg!

Metszéspontok x -koordinátáinak számítása

```
> metszéspontok := sort([solve(x^2 + (1 - x)^2 = 5/9, x)]);
a, b := metszéspontok_1, metszéspontok_2;
metszéspontok := [1/3, 2/3]
a, b := 1/3, 2/3
```

(1.3.1)

Tehát a két metszéspont $x_1 = \frac{1}{3}$ és $x_2 = \frac{2}{3}$.

A parabola az $\frac{5}{9}$ szint fölött van a $\left[0, \frac{1}{3}\right]$ és a $\left[\frac{2}{3}, 1\right]$ intervallumokon. Tehát az A halmaz két x -intervallum uniója

$$A = \left[0, \frac{1}{3}\right] \cup \left[\frac{2}{3}, 1\right].$$

Az A halmaz **mértéke** a két intervallum hosszának összege

$$m(A) = m\left(\left[0, \frac{1}{3}\right]\right) + m\left(\left[\frac{2}{3}, 1\right]\right) = \frac{1}{3} + \frac{1}{3} = \frac{2}{3}.$$

Ezért


$$P\left(\left\{x \in [0, 1] : x^2 + (1 - x)^2 \geq \frac{5}{9}\right\}\right) = \frac{m(A)}{m(\Omega)} = \frac{\frac{2}{3}}{1} = \frac{2}{3}.$$

Vagyis a keresett valószínűség $\frac{2}{3}$. Az alábbi rajz zöld vonallal szemlélteti az A eseményt az

$\Omega = [0, 1]$ intervallumon belül.

Az $x^2 + (1 - x)^2 > \frac{5}{9}$ pontok rajza a $[0, 1]$ intervallumban

```
> kedvező := plot([ [0, 0], [a, 0]], [[b, 0], [1, 0]], color = green, thickness = 5) :  
    vonalak := plot([ [a, 0], [a, 5/9]], [[b, 0], [b, 5/9]], color = black, linestyle = 2) :  
> plots:-display([rajz1, kedvező, vonalak])
```



(b) Igazoljuk a fenti számítás jogosságát a kísérlet **szimulációs** programjának megírásával!

Generálunk 500 darab x_n számot egyenletes eloszlással a $[0, 1]$ intervallumban. Majd rajzoljuk fel az


$$y_n = x_n^2 + (1 - x_n)^2 > \frac{5}{9}$$

feltételt teljesítő pontok relatív gyakoriságait a $P(A) = \frac{2}{3}$ valószínűséggel együtt.

Egyenletes eloszlású 500 pont generálása $[0, 1]$ intervallumban

```
> with(Statistics) : randomize( ) :  
> N := 500;  
> U := Sample(Uniform(0, 1), N) :  
    Histogram(U)
```

$N := 500$




Az $y = x^2 + (1 - x)^2 > \frac{5}{9}$ feltételt teljesítő pontok relatív gyakorisága
és $a p = \frac{2}{3}$ valószínűség

```

> Yn := map(x → x^2 + (1 - x)^2, U) :
> relativ_gyakoriságok := NULL : gyak := 0 :
  for k from 1 to N do
    if Yn[k] > 5/9 then gyak := gyak + 1 end if:

    relativ_gyakoriságok := relativ_gyakoriságok, gyak/k :
  end do:
  relativ_gyakoriságok := [relativ_gyakoriságok] :
> rajz1 := Statistics[LineChart](relativ_gyakoriságok, thickness = 3, axis = [gridlines = [10, color = black]], tickmarks = [10, 10], legend = [Az x^2 + (1 - x)^2 relatív gyakoriságai], view = [1..N, 0..1]) :
plots:-display([rajz1, plot(2/3, 1..N, color = red, thickness = 3)])

```



— Az $x^2 + (1 - x)^2$ relatív gyakoriságai

Az ábráról leolvasható, hogy az $y_n = x_n^2 + (1 - x_n)^2 > \frac{5}{9}$ feltételt teljesítő véletlen számok relatív gyakoriságai ingadoznak a $p = \frac{2}{3}$ valószínűség körül. Ezért az **elméleti** alapon történő számításaink összhangban vannak a szimulációk alapján kapott **gyakorlati** számításainkkal.

(c) Használjuk ki a *Statistics* csomag azon képességét, hogy folytonos valószínűségi változókon képes "egyszerűbb" műveleteket elvégezni! Így az X egyenletes eloszlásból képes az $Y = X^2 + (1 - X)^2$ művelettel kapott Y változóra a $P\left(Y > \frac{5}{9}\right)$ valószínűséget kiszámolni a *Probability* eljárás! Vagyis automatizálja mindazokat a tevékenységeket, amelyeket mi az előbb matematikai szempontból elvégeztünk!

Geometriai valószínűség véletlen változókkal

```
> X := RandomVariable(Uniform(0, 1))
> Y := X^2 + (1 - X)^2
> P(X^2 + (1 - X)^2 > 5/9) = Probability(Y > 5/9)
      X := _R3
      Y := _R3^2 + (1 - _R3)^2
      P(5/9 < X^2 + (1 - X)^2) = 2/3
```

(1.3.2)

Ugyanígy az Y változó sűrűség függvényét a $PDF(Y, x)$ vagyis a *ProbabilityDensityFunction* eljárás számolja az x helyen!


Y sűrűség függvénye és az $(y_n, n = 1 \dots 500)$ sorozat hisztogramja

```
> f := simplify(PDF(Y, x))
```

(1.3.3)

$$f := \begin{cases} 0 & x \leq \frac{1}{2} \\ \frac{1}{\sqrt{-1+2x}} & x < 1 \\ \frac{1}{2} & x = 1 \\ 0 & 1 < x \end{cases} \quad (1.3.3)$$

```
> suruseg := plot( f, x = 0..1.2, thickness = 3, color = red ) :
HY := Histogram(Yn) :
plots:-display( [HY, suruseg] )
```




[Láthatóan a sűrűség függvény jól illeszkedik a gyakoriság hisztogram értékekre!]

3.2.4. Valószínűségek számítása 2D-s geometriai eszközökkel

A következő példában **két véletlen számot** kell választani, de a számítások során csak az egyiket figyeljük. Ekkor az **eseménytér** a sík egy háromszöge lesz, az **esemény algebra** a háromszög mérhető tartományainak rendszere. Arra építünk a valószínűségek számításánál, hogy a **valószínűségi mérték** a tartományok **területével** arányos.

3.2.4.1. PÉLDA. 2D-s (Ω, S, P) valószínűségi mező nem téglalap eseménytérrel

Válasszunk véletlenszerűen egy $P(x, y)$ pontot az $A(1, 0)$, $B(0, 1)$ és $C(0, 0)$ csúcspontú H derékszögű háromszög belsejében (lásd az ábrát). A **kísérlet** kimenetele a választott P pont x -koordinátája.



C

A

(a) Írjunk **szimulációs programot** a kísérlet bemutatására úgy, hogy dupla véletlen pontokat generálunk **egyenletes eloszlással** az x és az y koordinátákra a $[0, 1]$ intervallumból. Majd **tartsuk meg** azokat az (x, y) pontokat, amelyek az ABC_{Δ} háromszögbe esnek. Rajzoljunk fel a háromszögbe eső 500 pontot és ezek x -koordinátáinak **gyakoriság hisztogramját**!

(b) Mekkora valószínűséggel esik a véletlen pont x -koordinátája a $[0.2, 0.5]$ intervallumba? Mutassuk meg, hogy az $f(x) = 2 - 2 \cdot x$ **sűrűség függvény** alatti terület segítségével számolhatjuk ki a

$$P(0.2 \leq x \leq 0.5)$$

valószínűséget!

(c) Rajzoljuk fel az x -koordináták $[0.2, 0.5]$ intervallumba esésének **relatív gyakoriságait** és a kiszámolt valószínűséget! Igazoljuk ezzel a (b) pont számításának helyességét!


Megoldás

Készítsünk ábrát a **kísérlet** bemutatására. A program írása rávezet bennünket a pontok helyes generálására!

A kísérlet bemutatása Maple-ben

```
> restart : with( plots ) : with( Statistics ) : randomize( ) :
H := polygonplot( [ [ 1, 0 ], [ 0, 1 ], [ 0, 0 ] ], color = yellow ) :
Pontx := Sample( Distribution( Uniform( 0, 1 ) ), 1 ) [ 1 ] :
Ponty := Sample( Distribution( Uniform( 0, 1 ) ), 1 ) [ 1 ] :
while Pontx + Ponty > 1 do
    Pontx := Sample( Distribution( Uniform( 0, 1 ) ), 1 ) [ 1 ] :
    Ponty := Sample( Distribution( Uniform( 0, 1 ) ), 1 ) [ 1 ] :
end do:
Pont := pointplot( [ Pontx, Ponty ] ) :
vonal := plot( [ [ Pontx, Ponty ], [ Pontx, 0 ] ], style = line, linestyle = 2, color
    = red, thickness = 3 ) :
S := textplot( [ Pontx, Ponty, a véletlen pont, align = { 'above', 'right' }, font'
```

```
= [ "times", "roman", 16 ] ) :  
display( [ H, Pont, S, vonal ], scaling = constrained )
```



(a) Miből táplálkozik a **kísérlet**? A generálásnál láttuk, hogy előbb az egységnégyzetben kell generálnunk egyenletes eloszlással pontokat. Majd ki kell válogatni azokat, amelyek a H háromszögbe esnek. Így a maradék pontok is egyenletes eloszlásban fordultak elő a H háromszögben! Tehát az **eseménytér** az $\Omega = \text{ABC}_{\Delta}$ háromszög belső pontjainak halmaza, amely

kettő dimenziós. Ennek mértéke $m(H) = \frac{1}{2}$ a **területe**.

Annak feltétele, hogy a $[0, 1] \times [0, 1]$ egységnégyzetben levő $P(x, y)$ pont az ABC háromszögben legyen az $x + y \leq 1$ feltétel teljesülése. Tegyük fel, hogy egyenletes eloszlással pontokat generáltunk a $H = \text{ABC}$ háromszög

$$H = \{ (x, y) \mid 0 \leq x \leq 1, 0 \leq y \leq 1 - x \}$$

belsejében. Világos, hogy az x -koordináták gyakrabban esnek $x = 0$ közelébe, mint $x = 1$ közelébe. Hiszen az $A(1, 0)$ csúcspont közelében csak kis területű tartományból választhatunk pontokat, ezzel ellentétben a BC függőleges oldal mentén sok olyan pont közül választhatunk, amelyek x -**koordinátája ugyanaz**!

Az alábbi program addig generál pontokat az egységnégyzetben, amíg 500 pontot nem kapunk a háromszög belsejében. Ezért nem tudjuk előre megmondani, hogy hányszor fut le a pontot generáló ciklus!


500 véletlen pont a háromszögben és a pontok x -koordinátáinak gyakoriság hisztogramja

```
> N := 500 :  
pontok := NULL : i := 0 :  
> while i ≤ N do  
  Px := Sample( Distribution( Uniform( 0, 1 ) ), 1 ) [ 1 ] :  
  Py := Sample( Distribution( Uniform( 0, 1 ) ), 1 ) [ 1 ] :  
  if Px + Py ≤ 1 then  
    pontok := pontok, [ Px, Py ] :  
    i := i + 1 :  
  end if:  
end do:
```

```

pontok := [ pontok ] :
rajzpontok := display( [ H, pointplot( pontok, title
= A pontok egyenletes eloszlással a háromszögben ) ] ) :
> pontokx := map( x → x[ 1 ], pontok ) : display( Matrix( 2, 1,
[ Histogram( pontokx, title
= Az x-koordináták gyakoriság histogramja ), rajzpontok ] ) )

```



(b) A histogramról leolvasható, hogy az x -koordináták nem egyenletes eloszlásban helyezkednek el a $[0, 1]$ intervallumban! A példa kérdése az, hogy milyen gyakran esnek az x -koordináták a $I = [0.2, 0.5]$ intervallumba!

Láttuk a 2.8. pontban, hogy az ilyen számításoknál fontos szerepet játszik egyrészt az

$$F(x) = P([0, x])$$

eloszlás függvény másrészt ennek deriváltja az

$$f(x) = \frac{d}{dx} F(x)$$

sűrűség függvény.

Ezek felhasználásával tetszőleges $[a, b] \subseteq [0, 1]$ részintervallum valószínűségét kétféleképpen is kiszámolhatjuk

$$P([a, b]) = F(b) - F(a) = \int_a^b f(x) dx$$

használva az integrálszámítás és a differenciálszámítás Newton-Leibniz-féle alap tételét.

Mondjuk meg az eseménytér ismeretében az $F(x) = P([0, x])$ valószínűséget tetszőleges $0 < x < 1$ mellett! A geometriai valószínűség számítási elve alapján

$$P([0, x]) = \frac{m(T_x)}{m(H)}.$$

A fenti tömör jelölés megtévesztő, mert a $[0, x]$ intervallum P valószínűségét nem a $[0, 1]$ eseménytérben kell számolni. A példa **alapfeltételezett** eseménytere a $H = ABC_{\Delta}$ derékszögű háromszög. Ebben a T_x halmaz


$$T_x = \{(a, b) \in H : 0 \leq a \leq x\}$$

olyan $P(a, b)$ pontokból áll, amelyek első koordinátája a $[0, x]$ intervallumba esik. Könnyű látni, hogy T_x egy trapéz, melynek csúcspontjai $C(0, 0)$, $B(0, 1)$, $D(x, 1 - x)$ és $E(x, 0)$. Rajzoljuk fel a trapézokat animációval az x értékeket növelve!

A trapézok animációja

```
> trapézok := [seq(polygonplot([ [0, 0], [0, 1], [x, 1 - x], [x, 0]], color = yellow), x = 0.1..1, 0.1)]:
> display(trapézok, insequence = true, scaling = constrained, title = A [0,x] intervallumon vett trapézok,
tickmarks = [10, 10])
```

A $[0, x]$ intervallumon vett trapézok



A trapéz területének képletét alkalmazva

$$m(T_x) = \frac{x \cdot (1 + 1 - x)}{2} = \frac{x \cdot (2 - x)}{2}$$

és így az eloszlás függvény

$$F(x) = P([0, x]) = \frac{\frac{x \cdot (2 - x)}{2}}{\frac{1}{2}} = x \cdot (2 - x), \text{ ha } 0 \leq x \leq 1.$$

A sűrűség függvényt az eloszlás függvényből x -szerint deriválással kapjuk, tehát



$$f(x) = \frac{d}{dx} F(x) = (x \cdot (2 - x))' = 2 - 2 \cdot x, \text{ ha } 0 \leq x \leq 1.$$

Az x -koordináták eloszlás függvénye

```
> x := 'x': F := x → x · (2 - x)
> plot(F(x), 'x' = 0..1, thickness = 3,
scaling = constrained)
F := x → x (2 - x)
```

Az x -koordináták sűrűség függvénye

```
> x := 'x': f := x → 2 - 2 · x
> sűrűség := plot(f(x), 'x' = 0..1,
thickness = 3, scaling
= constrained): sűrűség
f := x → 2 - 2 x
```



Tehát az x -koordináták $I = [0.2, 0.5]$ intervallumba esésének valószínűsége az eloszlás függvényével

$P([0.2, 0.5]) = F(0.5) - F(0.2) = 0.5 \cdot (2 - 0.5) - 0.2 \cdot (2 - 0.2) = 0.75 - 0.36 = 0.39$
és ugyanez a sűrűség függvényvel


$$P([0.2, 0.5]) = \int_{0.2}^{0.5} (2 - 2 \cdot x) dx = 0.39.$$

(c) A számítások elvi alapjait erősítsük meg egyrészt a gyakoriság hisztogram és a sűrűség függvény egymásra rajzolásával, másrészt a relatív gyakoriságok és a számolt valószínűség ábrájával.

A hisztogram, a sűrűség függvény és a trapéz terület

- > `befestett := plot(f(x), x = 0.2..0.5, filled = true, color = yellow) :`
- > `display([sűrűség, befestett, Histogram(pontokx)], tickmarks = [10, 10], title`
`= A hisztogram, a sűrűség függvény és a trapéz terület)`

A hisztogram, a sűrűség függvény és a trapéz terület




A sűrűség függvény jól illeszkedik a hisztogramhoz. Befestettük sárga színűre a sűrűség függvény $I = [0.2, 0.5]$ intervalluma alatti trapéz területet. Ez a terület szemlélteti a kapott $p = 0.39$ valószínűséget.

Most rajzoljuk fel a $0.2 \leq x \leq 0.5$ feltételt kielégítő pontok relatív gyakoriságainak változását a kiszámolt $p = 0.39$ valószínűséggel együtt!

Az x-koordináták relatív gyakoriságai és a valószínűség

```
> relativ_gyakorisagok := NULL : gyak := 0 :
for k from 1 to N do
  Px := pontok[k][1] :
  if (0.2 ≤ Px) and (Px ≤ 0.5) then gyak := gyak + 1 end if
  relativ_gyakorisagok := relativ_gyakorisagok,  $\frac{gyak}{k}$  :
end do
relativ_gyakorisagok := [relativ_gyakorisagok] :
> rajz1 := Statistics[LineChart](relativ_gyakorisagok, thickness = 3, axis = [gridlines = [10, color = black]], tickmarks = [10, 10], legend = ["Az x-koordináták relatív gyakoriságai"], view = [1..N, 0..1]) :
plots:-display([rajz1, plot(0.39, 1..N, color = red, thickness = 3)])
```



A relatív gyakoriságok a kiszámolt valószínűség körül ingadoznak és azokhoz egyre közelebb kerülnek! Tehát számításaink helyesek voltak.

3.2.5. Találkozási probléma és megoldása

3.2.5.1. PÉLDA

Rómeó és Júlia találkoztól beszélnek meg egy bizonyos órán belül. Ezen az egy órán belül bármelyik időpont egyformán valószínű mindkettőjük érkezésére. Aki elsőnek érkezik, az vár 15 percet. Ha a másik nem érkezik meg ezen a 15 percen belül, akkor elmegy és nem találkoznak.

(a) Mekkora valószínűséggel találkozik Rómeó és Júlia?

(b) Generáljunk 200 időpontot Rómeó érkezésére és 200 időpontot Júlia érkezésére egyenletes eloszlással! Rajzoljuk fel a találkozás esemény **relatív gyakoriságait**! Mutassuk meg, hogy a relatív gyakoriság értékek az (a) pontban kiszámolt valószínűség körül ingadoznak!

(c) A *Statistics* csomag *RandomVariable* és *Probability* eljárását használva adjuk meg az (a) pontban számolt valószínűséget!

Megoldás

(a) Jelölje X Rómeó érkezési időpontját az 1 órán belül és Y Júlia érkezési időpontját. Mivel X és Y egymástól függetlenül bárhol lehet az 1 órán belül, ezért az eseménytér

$$\Omega = \{(x, y) \mid 0 \leq x \leq 1, 0 \leq y \leq 1\} = [0, 1] \times [0, 1]$$

egységnégyzet. Az eseménytér kettő dimenziós, ezért $m(\Omega) = 1$ mértéke a **területe**.

A találkozás akkor jön létre, ha a két érkezés közötti eltérés nem nagyobb, mint $\frac{1}{4}$ óra = 15 perc. Ha Rómeó érkezik előbb, vagyis $X \leq Y$, akkor a **találkozáshoz** Júliának Rómeó érkezését követő $\frac{1}{4}$ órán belül meg kell érkezni. Egyenlőtlenségekkel felírva

$$(*) \quad X \leq Y \leq X + \frac{1}{4}.$$

Ha Júlia érkezik előbb, vagyis $Y \leq X$, akkor a **találkozáshoz** Rómeónak Júlia érkezését követő $\frac{1}{4}$ órán belül meg kell érkezni. Egyenlőtlenségekkel felírva

$$Y \leq X \leq Y + \frac{1}{4}.$$

Ez utóbbi egyenlőtlenség rendszert

(**)

$$X - \frac{1}{4} \leq Y \leq X$$

alakban írhatjuk.

A (*) és (**) egyenlőtlenségeket abszolút érték formában a következő tömör alakba fogjuk egybe

$$T = \left\{ (x, y) \in \Omega \mid |y - x| \leq \frac{1}{4} \right\}.$$

Rajzoljuk be a $[0, 1] \times [0, 1]$ egységnégyzetbe a T találkozás eseményének halmazát a *plots* csomag *inequal* eljárásával!

A találkozás T eseménye az eseménytérben


```
> restart : with(plots) :
```

```
> M := inequal( { x - 1/4 ≤ y, y ≤ x + 1/4 }, x = 0..1, y = 0..1, scaling = constrained, title
= A találkozás eseménye ) :
```

```
T := textplot( [0.5, 0.5, "T", align = { 'above', 'right' }, font = [ "times", "roman", 20 ] ] ) :
```

```
> display(M, T)
```

A találkozás eseménye



A találkozást a kék színű sáv szemlélteti, amely egy hatszög. A négyzet jobb alsó sarkában levő szürke pontokra a találkozás azért nem jön létre, mert Júlia korán jött (Y kicsi) és hozzá képest Rómeó túl későn érkezett (X nagy). A négyzet bal felső sarkában levő szürke pontokra a találkozás azért nem jön létre, mert Rómeó korán jött (X kicsi) és hozzá képest Júlia túl későn érkezett (Y nagy).

A hatszög területét úgy célszerű kiszámolni, hogy az egység négyzet területéből kivonjuk a kettő darab szürke derékszögű háromszög területének összegét. A két szürke háromszöget, ha egybe toljuk, akkor egy $a = \frac{3}{4}$ oldalhosszúságú négyzetet kapunk. Ezért a találkozás területe

$$m(T) = 1 - \left(\frac{3}{4} \right)^2 = 1 - \frac{9}{16} = \frac{7}{16}.$$

Tehát a találkozás valószínűsége

$$P(T) = P(\text{találkozás}) = \frac{m(T)}{m(\Omega)} = \frac{\frac{7}{16}}{1} = \frac{7}{16} = 0.4375.$$

Annak esélye, hogy Rómeó és Júlia találkozik 43.75%.


(b) Generálunk 200 véletlen számot a $[0,1]$ intervallumban egyenletes eloszlással Rómeó érkezésére és hasonlóan 200 értéket Júlia érkezésére! Ellenőrizzük, hogy a k -ik időpontra teljesül-e a találkozás

$$|x_k - y_k| < \frac{1}{4}$$

feltétele. Ezek relatív gyakoriságait felrajzoljuk a kiszámolt $p = \frac{7}{16}$ valószínűséggel együtt.

A találkozási esemény relatív gyakoriságainak szimulációja

```
> with(Statistics) : N := 200 :
X := Sample(Distribution(Uniform(0, 1)), N) : Y := Sample(Distribution(Uniform(0, 1)), N) :
> relativ_gyakoriságok := NULL : gyak := 0 :
for k from 1 to N do
  if  $|X_k - Y_k| < 0.25$  then gyak := gyak + 1 end if:
  relativ_gyakoriságok := relativ_gyakoriságok,  $\frac{\text{gyak}}{k}$  :
end do:
relativ_gyakoriságok := [relativ_gyakoriságok] :
> rajz1 := Statistics[LineChart](relativ_gyakoriságok, thickness = 3, axis = [gridlines = [10, color = black]], tickmarks = [20, 10], legend = ["A találkozás relatív gyakoriságai"], view = [1..N, 0..1]) :
plots:-display([rajz1, plot(0.4375, 1..N, color = red, thickness = 3)])
```



A találkozás eseményének relatív gyakoriságai a kiszámolt $p = 0.4375$ valószínűség körül ingadoznak.

(c) Definiáljuk az X és az Y érkezési időket, mint véletlen változókat, amelyek egyenletes eloszlásúak a $[0,1]$ intervallumban. Ezek szimbolizálják Rómeó és Júlia érkezési időpontjait.

Definiáljuk ezekkel a $T = |X - Y|$ eltérés változót! A találkozás eseményét a $\left\{T < \frac{1}{4}\right\}$

egyenlőtlenség írja le. Ennek $P\left(T < \frac{1}{4}\right)$ képlettel jelölt valószínűségét a *Statistics* csomag

$Probability\left(T < \frac{1}{4}\right)$ eljárása kiszámolja, áthidalva ezzel eddigi matematikai erőfeszítéseinket.

A találkozás valószínűsége véletlen változókkal

```
> with(Statistics) :
  X := RandomVariable(Uniform(0, 1)) :
  Y := RandomVariable(Uniform(0, 1)) :
  T := |X - Y| :
  A találkozás eseménye = Probability(T < 1/4)

  A találkozás eseménye = 7/16 (1.5.1)
```

A kapott valószínűség megegyezik az általunk kiszámolt értékkel.

3.2.6. Háromszög szerkeszthetőségi probléma és megoldása

3.2.6.1. PÉLDA.


Egy egységnyi hosszúságú pálcát véletlenszerűen választott 2 helyen eltörünk 3 kisebb darabra.

(a) Mekkora valószínűséggel szerkeszthető háromszög a keletkezett 3 szakaszból?

(b) Generáljunk 200 dupla töréspontot egyenletes eloszlással és rajzoljuk fel a háromszög szerkeszthetőség eseményének **relatív gyakoriságait**! Mutassuk meg, hogy a relatív gyakoriságok az (a) pontban kiszámolt valószínűség körül ingadoznak!

Megoldás

(a) A háromszög szerkeszthetőség eseményének egyértelmű leírásához jelölje a keletkezett három szakasz hosszát x , y és z .



Mivel a három szakasz összhossza 1, ezért $x + y + z = 1$.

Ebből kifejezhető valamelyik változó, legyen ez például z és így elegendő az x és y változó a törések helyeinek egyértelmű leírásához


$$z = 1 - x - y.$$

Egyértelmű, hogy az $0 \leq x \leq 1$ és $0 \leq y \leq 1$ egyenlőtlenségeknek teljesülni kell az oldalakra! Mivel azonban $z = 1 - x - y$ és $z \geq 0$, ezért az $x + y \leq 1$ egyenlőtlenségnek is fenn kell állni minden elemi eseményre. Ezek az egyenlőtlenségek egy háromszöget határoznak meg az $(x; y)$ Descartes-féle derékszögű koordináta-rendszer első negyedében

$$\Omega = \{(x, y) \mid 0 \leq x \leq 1, 0 \leq y \leq 1, x + y \leq 1\}$$

A pálca törésének eseménytere

```
> restart : with(plots) :
> eseménytér := polygonplot([ [0, 0], [1, 0], [0, 1] ], color = "DarkGrey", thickness = 3,
  scaling = constrained, tickmarks = [10, 10]) : eseménytér
```



Ellenőrizzük az ilyen probléma esetén, hogy a feladat által leírt **kísérlet kimenetelei** és az **eseménytér elemei** között kölcsönösen egyértelmű megfeleltetést tudunk-e létrehozni. Tehát bármely (x, y, z) és $x + y + z = 1$ töréshez egyértelműen hozzárendelhető-e egy (x, y) pont az Ω háromszögben. Fordítva, bármely (x, y) ponthoz az Ω háromszögből egyértelműen hozzárendelhető-e egy (x, y, z) és $x + y + z = 1$ törés.

Tehát az eseménytér kettő dimenziós és ennek $m(\Omega) = \frac{1}{2}$ mértéke a **területe**.

A háromszög szerkeszthetőség feltétele az úgynevezett **háromszög egyenlőtlenségek** teljesülése. Melyszerint bármely két oldal összege legyen nagyobb, mint a harmadik oldal. Ez 3 egyenlőtlenség felírását jelenti

$$H = \{ (x, y) \in \Omega \mid x + y \geq z, x + z \geq y, y + z \geq x \}.$$

A H halmaz egyenlőtlenségeibe helyettesítsük be a $z = 1 - x - y$ kifejezést és rendezzük.


Feltételek a háromszög szerkeszthetőséghez		
$x + y \geq z$	$x + z \geq y$	$y + z \geq x$
$x + y \geq 1 - x - y$	$x + (1 - x - y) \geq y$	$y + (1 - x - y) \geq x$
$x + y \geq \frac{1}{2}$	$\frac{1}{2} \geq y$	$\frac{1}{2} \geq x$

A kapott határoló egyeneseket rajzoljuk be az eseménytérbe és adjuk meg az általuk kijelölt tartományt!

A háromszög szerkeszthetőség H eseménye az eseménytérben

```
> H := polygonplot( [ [ [ 1/2, 0 ], [ 1/2, 1/2 ], [ 0, 1/2 ] ], color = "Red", thickness = 3 ] :
T := textplot( [ 0.3, 0.3, "H", align = { 'above', 'right' }, font = [ "times", "roman", 20 ] ] ) :
display( [ H, eseménytér, T ], title = Háromszög szerkeszthetőség eseménye a piros háromszög)
```

Háromszög szerkeszthetőség eseménye a piros háromszög



A három szürke és a piros háromszög egybevágóak, ezért 4 egyenlő területű részt kaptunk az eseménytérben.

A fenti ábra piros színnel jelölt H háromszögéből kiválasztva egy (x, y) pontot, olyan $z = 1 - x - y$ értéket kapunk, amellyel az (x, y, z) hosszakból háromszög szerkeszthető. Mivel a H halmaz egy egyenlőszárú derékszögű háromszög, melynek befogója $\frac{1}{2}$ hosszú, ezért területe

$$m(H) = \frac{\frac{1}{2} \cdot \frac{1}{2}}{2} = \frac{1}{8}.$$

Tehát a háromszög szerkeszthetőség valószínűsége

$$P(H) = \frac{\frac{1}{8}}{\frac{1}{2}} = \frac{1}{4}.$$

Azt mondhatjuk, hogy átlagosan minden négy véletlenszerű törésből egy olyan lesz, amelyből háromszög szerkeszthető.

(b) Az egységnégyzetben egyenletes eloszlással addig képezünk (x, y) pontokat, amíg az eseménytérben 200 pontot nem kapunk. Az eseménytér háromszögéhez hez nem tartozó pontokat eldobjuk és a többi $(x, y, 1 - x - y)$ formában gyűjtjük. A pontokat berajzoljuk az eseménytérbe, hogy vizuálisan ellenőrizni tudjuk az egyenletes eloszlás feltételének teljesülését.


200 véletlen pont generálása egyenletes eloszlással az eseménytérben

```
> with(Statistics) : randomize( ) :
N := 200 :
törések := NULL : i := 0 :
> while i ≤ N do
    x := Sample(Distribution(Uniform(0, 1)), 1)[1] :
    y := Sample(Distribution(Uniform(0, 1)), 1)[1] :
    if x + y < 1 then
```

```

z := 1 - x - y :
törések := törések [ x, y, z ] :
i := i + 1 :
end if:
end do:
törések := [ törések ] :
> display( [ pointplot( map( a → [ a1, a2 ], törések ), eseménytér ) ] )

```




A háromszög szerkeszthetőség relatív gyakoriságainak szimulációja

```

> relativ_gyakoriságok := NULL : gyak := 0 :
for k from 1 to N do
  x := törések[k][1] : y := törések[k][2] : z := törések[k][3] :
  if (x + y > z) and (x + z > y) and (y + z > x) then gyak := gyak + 1 end if:
  relativ_gyakoriságok := relativ_gyakoriságok,  $\frac{gyak}{k}$  :
end do:
relativ_gyakoriságok := [ relativ_gyakoriságok ] :
> rajz1 := Statistics[LineChart](relativ_gyakoriságok, thickness = 3, axis = [ gridlines = [ 10, color = black ], tickmarks = [ 20, 10 ], legend = [ "A háromszög szerkeszthetőség relatív gyakoriságai", view = [ 1..N, 0..1 ] ] :
plots:-display( [ rajz1, plot(0.25, 1..N, color = red, thickness = 3) ] )

```



Láthatóan a relatív gyakoriságok a $p = 0.25$ valószínűség körül ingadoznak.

3.2.7. Buffon-féle tű probléma és megoldása

A probléma eseményterét megtalálni a következő problémánál nem egyszerű! Továbbá a kedvező esetek mértékének kiszámításához az integrálszámítást is fel kell használni.

3.2.7.1. PÉLDA

Tekintsünk egy sík felületű játékasztalt, melynek felületére **párhuzamos egyeneseket** rajzoltunk egymástól **egységnyi távolságra**. Rádobunk az asztallapra egy **egységnyi hosszúságú tűt** véletlenszerűen többször egymásután és feljegyezzük, hogy a tű az asztallap valamelyik egyenesét metszi-e vagy sem.

Határozzuk meg annak valószínűségét, hogy a tűvel **eltalálunk egy egyenest?**

Megoldás

A lehetséges kimenetek egyértelmű megadásához hány független változó szükséges?

Néhány véletlen kísérlet	A tű pozíciójának leírása
	$0 \leq d \leq \frac{1}{2}, 0 \leq \varphi \leq \frac{\pi}{2}$

A lehetséges elemi kimeneteket a következő két koordinátával tudjuk leírni

- Jelölje d a **tű középpontjának** távolságát a **legközelebbi egyenestől**.
- Jelölje φ a **tű és az egyenesek hajlásszögét** radiánban mérve.

A d **távolság** értéke 0 és $\frac{1}{2}$ között változhat. Ha ugyanis $d > \frac{1}{2}$ lenne, akkor már egy másik egyeneshez lesz közelebb a tű közepe, mint $\frac{1}{2}$. Ezért d értékére válasszuk ezt az $\frac{1}{2}$ -nél kisebb távolságot.

A φ **szög** értéke 0 és

$\frac{\pi}{2}$ között változhat. Ha ugyanis $\varphi > \frac{\pi}{2}$ lenne, akkor a tű és a


párhuzamosok közötti két szög közül válasszuk φ értékére a hegyesszöget.

Így a leesett tű helyzetét egyértelműen jellemeztük az egyenesekhez viszonyítva a (φ, d) számpárral, ezért az eseménytér

$$\Omega = \left\{ (\varphi, d) \mid 0 \leq \varphi \leq \frac{\pi}{2}, 0 \leq d \leq \frac{1}{2} \right\}$$

Az eseménytér egy téglalap, melynek területe

$$m(\Omega) = \frac{\pi}{2} \cdot \frac{1}{2} = \frac{\pi}{4}.$$



A jobb oldali ábra jelölései:

- O a tű középpontja
- P a tű egyenesének metszéspontja az O ponthoz legközelebbi egyenessel
- B az O pont merőleges vetülete az O ponthoz legközelebbi egyenesen
- d jelöli az O és B pontok távolsága: $d = |OB|$

Megállapítás

A tű akkor és csakis akkor metsz egy egyenest, ha teljesül az $|OP| \leq \frac{1}{2}$ egyenlőtlenség.

Az OBP_{Δ} derékszögű háromszögből meghatározzuk az OP távolságot d és φ segítségével

$$\sin(\varphi) = \frac{d}{|OP|}$$

$$|OP| = \frac{d}{\sin(\varphi)}$$

Tehát a tű pontosan akkor metszi valamelyik egyenest, ha teljesül a

$$\frac{d}{\sin(\varphi)} \leq \frac{1}{2}$$

egyenlőtlenség.

A metszésre nézve kedvező eseteket az

$$M = \left\{ (\varphi, d) \mid d \leq \frac{1}{2} \cdot \sin(\varphi) \right\}$$


tartomány adja meg.

Tehát ha a

$d = \frac{1}{2} \cdot \sin(\varphi)$ függvényt berajzoljuk a (φ, d) koordinátarendszerbe, akkor ezen görbe grafikonja alatti tartomány a kedvező eseteket tartalmazza az Ω eseménytérén belül..

Az eseménytér és benne a tű és egyenes metszésének eseménye

```
> restart :
with(plots) :
eseménytér := polygonplot( [ [0, 0, ], [  $\frac{\pi}{2}$ , 0 ], [  $\frac{\pi}{2}$ ,  $\frac{1}{2}$  ], [ 0,  $\frac{1}{2}$  ] ], color = grey ) :
> M := plot(  $\frac{1}{2} \cdot \sin(\varphi)$ ,  $\varphi = 0 .. \frac{\pi}{2}$ ,  $d = 0 .. 0.5$ , filled = true, color = "DarkSlateGray", tickmarks
= [piticks, 5] ) :
T := textplot( [0.4, 0.1, A tű és egy egyenes metszési eseménye, align= { 'above', 'right' }, font'
= [ "times", "roman", 14 ] ] ) :
display( [M, eseménytér, T] )
```



Az M eseményt ábrázoló tartomány területét határozott integrállal tudjuk kiszámítani

$$m(M) = \int_0^{\frac{\pi}{2}} \frac{1}{2} \cdot \sin(\varphi) \, d\varphi = \frac{1}{2} \cdot [-\cos(\varphi)]_{\varphi=0}^{\varphi=\frac{\pi}{2}} = \frac{1}{2} \left(-\cos\left(\frac{\pi}{2}\right) + \cos(0) \right) = \frac{1}{2}.$$

Tehát a tű és az egyenesek metszésének valószínűsége

$$P(M) = \frac{m(M)}{m(\Omega)} = \frac{\frac{1}{2}}{\frac{\pi}{4}} = \frac{2}{\pi} \approx 0.6366.$$

Tehát kb. 63,7% annak esélye, hogy a tűvel eltaláljuk a párhuzamos egyenesek valamelyikét.

Elfogadva azt az elvet, miszerint egy valószínűséget a relatív gyakoriságok egyre jobban megközelítenek, ha a kísérletek száma nagy, akkor ebből kiszámíthatjuk π értékét közelítőleg a

$$\pi = \frac{2}{P(M)} \approx \frac{2 \cdot n}{k_n(M)}$$

képlet alapján, ahol $k_n(M)$ jelöli azt, hogy n kísérletből hányszor következett be az M metszési esemény.

Erre felépítve több kísérletet is elvégeztek π értékének közelítő kiszámítására. A következő táblázat tartalmazza ezen kísérletek főbb adatait: a kísérletező személyt, a tű hosszát, a dobások számát, a találatok számát és π értékére kapott közelítő értéket.

Kísérletező	A tű hossza	Dobások száma	Találatok száma	Becslés π -re
Wolf, 1850	.8	5000	2532	3.1596
Smith, 1855	.6	3204	1218	3.1553
De Morgan, c.1860	1.0	600	382	3.137
Fox, 1864	.75	1030	489	3.1595
Lazzerini, 1901	.83	3408	1808	3.1415929
Reina, 1925	.5419	2520	869	3.1795

Az alábbi internet címen interaktív programot találhatunk, melyben a dobálásokkal szemléltetik π közelítő értékének számítását: <http://vis.supstat.com/2013/04/buffons-needle/>

Látható, hogy a módszer nagyon lassan konvergál, ezért a gyakorlatban nem alkalmas π értékének nagy pontosságú számítására.

3.2.8. Bertran-féle paradoxon levezetése

A következő példa azt mutatja, hogy az eseménytér geometriai reprezentációja meghatározza a valószínűség értékét! Különböző reprezentációk mellett más és más valószínűségeket kaphatunk ugyanarra a problémára! Az ellentmondás vagy paradoxon csak látszólagos és az eseménytér reprezentációjával válik egyértelművé a valószínűség meghatározása.

PROBLÉMA

Válasszuk ki véletlenszerűen egy adott **kör** valamelyik **húrját** és számítsuk ki annak valószínűségét, hogy a kiválasztott húr nagyobb, mint a körbe írható **szabályos háromszög oldala**.

Megoldás


A keresett valószínűség értéke attól függően más és más lesz, hogy milyen koordinátát használunk a húrok azonosítására. A véletlen szó jelentése is hozzájárul ahhoz, hogy a megoldásra kapott **három** különböző érték mindegyike helyes lesz. Az egyértelmű eredményhez a feladat szövegét pontosítani kell!

Az Ω eseménytér a körben kiválasztható különböző húrok lesznek. Azonban valahogyan számszerűsíteni kell a különböző húrokat úgy, hogy koordinátákat kell egyértelműen hozzárendelni és ez megtehető többféle módon is.

Feltehető, hogy a kör sugara $r = 1$ és a koordináta-rendszer origóját a kör középpontjába helyeztük!

Jelölje A és B a húr két végpontját és M a középpontját. Legyen továbbá

α az A pont, β a B pont és θ az M pont polárszöge a koordinátarendszerben!



Az AB húr egyértelműen leírható az alábbi 3 különböző módon

(1) Az M pont (x,y) derékszögű koordinátaival

(2) Az M pont (r,θ) polár - koordinátaival

(3) Az A és B pontok $(1,\alpha)$ és $(1,\beta)$ polár - koordinátaival

Nézzük, hogyan lehet az AB húr L hosszát kiszámítani a fenti reprezentációk mellett!

A **véletlen** szó mindhárom esetben azt jelenti, hogy a megadáshoz használt koordinátákat véletlenszerűen választjuk. A feladat kérdése arra vonatkozik, hogy mikor lesz $L > \sqrt{3}$.

1. megoldási módszer

A húr $M(x, y)$ középpontjának a kör belsejében kell lenni, azaz

$$x^2 + y^2 \leq 1$$

egyenlőtlenségnek teljesülni kell! Az M pont megadásával a húr egyértelműen megrajzolhatjuk és ez fordítva is igaz!

A húr L hosszát x és y-nal az $L = 2 \cdot \sqrt{1 - (x^2 + y^2)}$ képlet adja meg, mert Pitagorász-tétele alapján

$$\begin{aligned} |OM| &= \sqrt{x^2 + y^2} \\ |AM| &= \sqrt{OP^2 - OM^2} = \sqrt{1 - (x^2 + y^2)} \\ L &= |AB| = 2 \cdot |AM| = 2 \cdot \sqrt{1 - (x^2 + y^2)}. \end{aligned}$$

Ha az $M(x, y)$ pont a szabályos háromszögbe írt kör belsejében van, akkor a húr hosszabb, mint a szabályos háromszög oldala. Ennek a körnek a sugara $r = \frac{1}{2}$. Ezért a összes esetek halmaza az $R=1$ sugarú kör


$$\Omega = \{ (x, y) \mid x^2 + y^2 \leq 1 \}.$$

A kedvező esetek halmaza

$$A = \left\{ (x, y) \mid x^2 + y^2 \leq \left(\frac{1}{2} \right)^2 \right\}.$$

Ezért a valószínűség

$$p = \frac{\Omega \text{ területe}}{A \text{ területe}} = \frac{r^2 \cdot \pi}{1^2 \cdot \pi} = \left(\frac{1}{2} \right)^2 = \frac{1}{4}.$$



2. megoldási módszer

A húr M középpontja megadható egyértelműen a kör O középpontjától mért $r = |OM|$ távolsággal és az $|OM|$ szakasz x-tengely pozitív felével bezárt θ szöggel. A feltételek ekkor

$$0 \leq r \leq 1 \text{ és } 0 \leq \theta \leq 2 \cdot \pi.$$

A húr L hosszát r segítségével a $L = 2 \cdot \sqrt{1 - r^2}$ képlet adja meg, mert $r^2 = x^2 + y^2$ az előzőek alapján.

Tehát az L hossz nem függ az M pont θ polárszögétől. Így kiválasztva egy húrt és azt forgatva a középpont körül megkapjuk az összes ugyanolyan hosszúságú húrt. Ezért


$$\Omega = \{ r \mid 0 \leq r \leq 1 \} = [0, 1].$$

Ennek megfelelően a kedvező események halmaza

$$A = \{ r \mid 0 \leq r \leq \frac{1}{2} \}.$$

Ezért a valószínűség

$$p = \frac{\left[0, \frac{1}{2}\right] \text{ intervallum hossza}}{[0, 1] \text{ intervallum hossza}} = \frac{\frac{1}{2}}{1} = \frac{1}{2}$$



3. megoldási módszer

A szimmetria miatt az A és B végpontok közül az egyiket, mondjuk B-t tetszőlegesen rögzíthetjük. Legyen B az (1,0) koordinátájú pont a vízszintes tengelyen. Az A pontot válasszuk véletlenszerűen a körvonalon bárhol. Az A pont $(1, \alpha)$ polárkoordinátáira a

$$0 \leq \alpha \leq 2 \cdot \pi$$

egyenlőtlenség teljesül.

A húr L hosszát a cos-tétel segítségével is megadhatjuk

$$L = \sqrt{2 - 2 \cdot \cos(\alpha)},$$

Tehát az eseménytér


$$\Omega = \{ \alpha \mid 0 \leq \alpha \leq 2 \pi \} = [0, 2 \pi].$$

Az AB húr hossza akkor lesz nagyobb mint a szabályos háromszög oldala, ha az α szög a $2\pi/3$ és $4\pi/3$ értékek között változik. Ennek megfelelően a kedvező események halmaza

$$A = \{ \alpha \mid \frac{2 \pi}{3} \leq \alpha \leq \frac{4 \pi}{3} \} = \left[\frac{2 \pi}{3}, \frac{4 \pi}{3} \right].$$

Ezért a valószínűség

$$p = \frac{\left[\frac{2 \pi}{3}, \frac{4 \pi}{3} \right] \text{ hossza}}{[0, 2 \pi] \text{ hossza}} = \frac{\frac{2 \pi}{3}}{2 \pi} = \frac{1}{3}$$



A három megoldás mindegyike helyes, tehát ugyanarra a problémára 3 különböző eredményt is kaphatunk. A kapott eredmények $\frac{1}{4}$, $\frac{1}{2}$ és $\frac{1}{3}$. Az eltérések onnan adódnak, hogy a feladatot nem fogalmaztuk meg pontosan. Vagyis az ellentmondás csak látszólagos.

Tanulság. Ha valószínűségi problémákat fogalmazunk meg, akkor úgy kell azt tennünk, hogy

La megoldás egyértelmű legyen!

3.2.9. Elméleti ellenőrző kérdések

3.2.9.1. Milyen mező esetén használjuk a valószínűségek számításának geometriai módszerét? Mondjon rá példát!

3.2.9.2. Az 1 dimenziós eseménytérnél hogyan számoljuk ki egy A esemény $P(A)$ valószínűségét? Mondjon rá példát!

3.2.9.3. A 2 dimenziós eseménytérnél hogyan számoljuk ki egy A esemény $P(A)$ valószínűségét? Mondjon rá példát!

3.2.9.4. Milyen alapelveket használ a Rómeó és Júlia találkozási példa?

3.2.9.5. Mekkora a valószínűsége a háromszög szerkeszthetőségnek egy véletlen pálca három részre törésekor?

3.2.9.6. A Buffon tű dobálási kísérleténél mik lesznek azok a változók, amellyel egyértelműen leírhatunk egy kimenetelt?

3.2.9.7. Honnan eredete az ellentmondás a Bertran-féle problémánál?

3.2.10. Gyakorló feladatok

3.2.10.1. feladat

Hajótöröttek egy lakatlan szigeten azt tervezik, hogy a viharban összeroncsolódott hajójuk árboc rúdjának felhasználásával új, kisebb hajót építenek. A vihar a 40 méteres árbocrudat három véletlenszerű darabra törte szét. Ha van olyan darab, amelyik 20 méternél hosszabb, akkor a kisebb hajót meg tudják építeni. Mekkora valószínűséggel találunk olyan hosszú darabot, hogy a hajó megépíthető legyen, amikor kiúsznak a roncsokhoz?

3.2.10.2. feladat

Egy kikötőbe két hajó érkezését várják. A hajók érkezése egymástól független és egy 24 órás időtartamon belül bármikor bekövetkezhet. Az egyik hajó kirakodása 2 órahosszat, a másik kirakodása 4 órahosszat vesz igénybe. Amelyik hajó előbb érkezik, azt elkezdik kirakodni és a másiknak várakozni kell a rakodás végéig, ha közben érkezik.

Mekkora valószínűséggel nem kell várakozni egyik hajónak sem?

3.2.10.3. feladat

Az „a” és „b” számokat véletlenszerűen és egymástól függetlenül választjuk a $[0,1]$ intervallumban.

Mekkora a valószínűsége, hogy az $a \cdot x^2 + 2 \cdot b \cdot x + 1 = 0$ másodfokú egyenlet gyökei komplex számok?

Rajzolja fel az eseményteret és a kedvező esetek halmazát!

3.2.10.4. feladat

Az „a” és „c” számokat véletlenszerűen és egymástól függetlenül választjuk a $[0,1]$ intervallumban.

Mekkora a valószínűsége, hogy az $a \cdot x^2 + x + c = 0$ másodfokú egyenlet gyökei valós számok? Rajzolja fel az eseményteret és a kedvező eseteket!

3.2.10.5. feladat

A $[0,1]$ intervallumban véletlenszerűen és egymástól függetlenül választjuk az x és y számokat. Mekkora a valószínűsége, hogy az $\left|y - \frac{1}{2}\right| + x > \frac{1}{3}$ feltétel teljesül, vagyis y távolsága $\frac{1}{2}$ -től és x összege nagyobb $\frac{1}{3}$ -nál?

3.2.10.6. feladat

A $[0,1]$ intervallumban véletlenszerűen és egymástól függetlenül választjuk az x és y számokat. Mekkora a valószínűsége, hogy az $\left|y - \frac{1}{2}\right| + x < \frac{1}{4}$ feltétel teljesül, vagyis y távolsága $\frac{1}{2}$ -től és x összege kisebb $\frac{1}{4}$ -nél?

3.2.10.7. feladat

A $[0, 1] \times [0, 1]$ egység négyzetben véletlenszerűen választunk egy $P(x, y)$ pontot

$(0 \leq x \leq 1, 0 \leq y \leq 1)$. Mekkora a valószínűsége, hogy a pont koordinátáira teljesül az $x \cdot y \leq \frac{1}{4}$

egyenlőtlenség?

3.2.10.8. feladat

A $[0, 1] \times [0, 1]$ egység négyzetben véletlenszerűen választunk egy $P(x, y)$ pontot

$(0 \leq x \leq 1, 0 \leq y \leq 1)$. Mekkora a valószínűsége, hogy a pont koordinátáira teljesül az $x \cdot y \geq \frac{1}{2}$

egyenlőtlenség?

3.2.10.9. feladat

Egy $a = 2$ oldalhosszúságú négyzet belsejében kiválasztunk véletlenszerűen egy pontot. Mennyi a valószínűsége, hogy a pont közelebb van valamelyik csúcshoz, mint a középponthoz?

3.2.10.10. feladat

Egy ötemeletes házban az egyes emeletek között 6 m távolság van, a földszint és az első emelet között 8m. Ha a liftajtó 2m, akkor mennyi a valószínűsége annak, hogy a lift véletlenszerű megakadásakor az ajtót teljes egészében fal takarja?

3.2.10.11. feladat

Az $A(0,0)$, $B(2,0)$, $C(2,2)$ és $D(0,2)$ csúcspontú egység négyzetben kiválasztunk véletlenszerűen egy $P(x, y)$ pontot. Mennyi a valószínűsége, hogy a P pont az AB oldalhoz közelebb van, mint a négyzet $O(1,1)$ középpontjához?

3.2.10.12. feladat

Kör alakú R sugarú céltáblára lövünk. (A találat valószínűsége arányos a területtel.) A céltáblát koncentrikus körökkel 10 részre akarjuk felosztani úgy, hogy mind a 10 részben a találat valószínűsége ugyanakkora legyen. Hogyan kell a körök sugarait megválasztani?

3.2.10.13. feladat

Válasszunk két véletlen számot x -t és y -t a $[0, 1]$ intervallumból. Mekkora valószínűséggel lesz a két szám egymáshoz 0.5-nél közelebb?

3.2.10.14. feladat

Egy egységnyi oldalhosszúságú négyzetre véletlenszerűen rádobunk egy r sugarú körlapot $\left(r < \frac{1}{2}\right)$.

Mekkora valószínűséggel esik a körlap teljes egészében a négyzet belsejébe?

3.2.10.15. feladat

A $[0,1]$ intervallumban egymástól függetlenül és véletlenszerűen választunk egy X és egy Y számot. Határozzuk meg annak valószínűségét, hogy az egyik szám kisebb lesz, mint a másik kétszerese!

3.2.10.16. feladat

A $[0,1]$ intervallumban véletlenszerűen választunk egy A számot $(0 \leq A \leq 1)$. Jelölje X a kapott A és $(1 - A)$ intervallumok közül a hosszabbikat, azaz $X = \max(A, 1 - A)$.

(a) Adja meg a $P\left(X < \frac{3}{4}\right)$ valószínűséget!

(b) Számolja ki a $P\left(X > \frac{4}{5}\right)$ valószínűséget!

